
QU I C K  R E F E R E N C E  ·  TK94

Five-Direction Joystick
Four tilts and a press on one analog pin, each its own fraction of VCC.

VCC is your board's logic voltage

A push to the right joins SIGNAL straight to VCC, so whatever VCC is
reaches your pin. On an ESP32, ESP32-S3 or Pico wire VCC to 3V3,
never 5V. On an Uno wire it to 5V: the sketch's lines assume VCC is
what the ADC measures against.

Pins, and which to wire
Parts up, header at the bottom. Wire three: GND, VCC, and SIGNAL to an analog pin. NC goes nowhere.
The names below are for the block turned a quarter, header on the left.

1 2 3 4

Front GND VCC NC SIGNAL

Read it GND VCC SIGNAL

Five-way switch, black cap
Four tilts and a press,
each a different voltage.

100 nF on SIGNAL
Letting go takes about
3 ms to fall to 0.

GND is the square pad
Count from it. The back
prints no pin names.

Five 10 kΩ in a chain
VCC to GND. Each push

joins SIGNAL to one point.

10 kΩ, SIGNAL to GND
It rests at 0 V,

not in the middle.

NC goes nowhere
No trace leaves its pad.

Leave it empty.

One analog wire, not five switches SIGNAL is 0 V with nothing pressed. Each push joins it to its own point of a
resistor ladder: up 0.37 V, press 0.60, left 0.90, down 1.47, right 3.30 at 3.3 V.
Read it with analogRead; a digital pin sees only right.

Six levels, and room between them

At rest

0 V: 10 kΩ holds
SIGNAL at GND

Not the middle of the scale. A reading
near 0 is nothing pressed.

Five pushes

5 levels, as
fractions of VCC

Up 0.11

Press 0.18

Left 0.27

Down 0.44

Right 1.00

Tightest pair

233 mV, up to press,
at 3.3 V

72 counts on an Uno. The line halfway
leaves half of it either side.

A sketch draws a line halfway between each pair of levels and names a reading by the lines it is above: 722, 359, 227, 146, 56
thousandths of VCC. 1 % resistors move a level by half a percent at most, far inside the room. Up and left held together read 200:
a press.

Levels worked out from the ladder: five 10 kΩ from VCC to GND, a push joining SIGNAL to one join, 10 kΩ and 100 nF from SIGNAL to GND. They agree with the maker's
own measured Uno readings. The switch is XUNPU's WS-1001-ACW09026: 250 ± 50 gf, 0.2 mm travel, 50,000 cycles. The chain draws 66 µA at 3.3 V. The sketch
compiles for an ESP32-S3, an ESP32 and an Uno.

The first read, on an ESP32-S3
Arduino IDE · Board: ESP32S3 Dev Module · USB CDC On Boot: Enabled · No library

// Uno: A0. ESP32: 34. ESP32-S3: 4. Pico: 26.

const int SIGNAL_PIN = 4;

// Highest level first; the line under each, thousandths of VCC.

const char* NAME[] = {"right", "down", "left", "press", "up"};

const int LIMIT[] = {722, 359, 227, 146, 56};

int readPermille() {

#if defined(ARDUINO_ARCH_ESP32)

  return analogReadMilliVolts(SIGNAL_PIN) * 1000L / 3300;

#else

  return analogRead(SIGNAL_PIN) * 1000L / 1023;

#endif

}

int whichPush(int permille) {

  for (int i = 0; i < 5; i++) {

    if (permille > LIMIT[i]) return i;

  }

  return -1;

}

void setup() {

  Serial.begin(115200);

}

void loop() {

  int p = readPermille();

  int d = whichPush(p);

  Serial.print("SIGNAL ");

  Serial.print(p);

  Serial.print("/1000  ");

  Serial.println(d < 0 ? "none" : NAME[d]);

  delay(200);

}

SIGNAL 0/1000 none

SIGNAL 444/1000 down

Serial Monitor at 115200, five lines a second. Right prints a little under 1000 on an
ESP32, the top of its ADC.

Wiring
FRONT S3 UNO ESP32 PICO

1 GND GND GND GND GND

2 VCC 3V3 5V 3V3 3V3

3 NC

4 SIGNAL GPIO 4 A0 GPIO 34 GP26

Hole 1 is the square pad. VCC is each board's own logic
voltage: 3V3 on the ESP32 boards and the Pico, 5V on the
Uno. SIGNAL goes to an ADC pin; on an ESP32 an ADC1 pin,
which keeps reading with Wi-Fi on.

One push, one step
Letting go, SIGNAL falls through every lower level on its way to 0: about 3 ms, as 100 nF drains through 10 kΩ. A sketch that acts on every reading
sees extra pushes. Believe a push only after it has read the same for 10 ms, and act only when it changes: the handbook's menu sketch does both.

When it does not work
Always reads 0
VCC not connected, or SIGNAL on NC, the third hole.

Only right does anything
SIGNAL read with digitalRead, or on a pin with no ADC.

Pushes named as lower ones
VCC is not what the ADC measures against. Uno: 5V.

Extra moves on letting go
The fall after a release. Wait 10 ms for a steady reading.

A press nobody made
Up and left held together read as a press.

Up reads as right
Held header-down. The names are for header on the left.

Specifications
Switch XUNPU WS-1001-ACW09026, five-way Output One analog line, 0 V at rest

Levels 1/9, 2/11, 3/11, 4/9 and all of VCC Lines 722, 359, 227, 146, 56 thousandths

Ladder 5 × 10 kΩ 1 %, 10 kΩ and 100 nF to GND Current 66 µA at 3.3 V; 0.4 mA holding right

Supply Your board's logic voltage on VCC Height 12.8 mm, board bottom to cap

TK94 Five-Direction Joystick 2 / 2


